
1 Executive Summary

AT A GLANCE

THREAT ACTOR

INC Ransom (GOLD IONIC) — Ransomware-as-a-Service

PRIMARY TARGETS

Healthcare, Education, Government, Manufacturing, Critical

Infrastructure

ATTACK VECTOR

Exploitation of public-facing appliances (Citrix, Fortinet),

Gootloader via SEO poisoning

BUSINESS IMPACT

Double extortion: data theft + encryption; 300+ victims published

in 2025; most deployed ransomware in July 2025

INC Ransomware is a Ransomware-as-a-Service (RaaS) operation first observed in July 2023. Tracked by Secureworks as GOLD IONIC and

cataloged by MITRE as G1032, the group employs a double-extortion model combining data theft with file encryption. INC operators have

claimed over 300 victims in 2025 alone, with healthcare as the most impacted sector according to MOXFIVE incident response data.

The group gains initial access primarily through exploitation of internet-facing appliances, notably Citrix NetScaler (CVE-2023-3519, CVE-

2023-4966) and Fortinet products (CVE-2023-48788, FG-IR-24-535). Post-compromise, operators deploy a combination of legitimate

remote management tools (AnyDesk, ScreenConnect, TightVNC) and offensive tooling (Impacket, lsassy.py, Meterpreter) for credential

harvesting, lateral movement, and data exfiltration via MEGASync, Rclone, and Restic.

This guide provides 10 structured hunt modules to systematically detect INC Ransom pre-encryption activity across your environment, from

initial access indicators through data exfiltration to ransomware execution artifacts. Typical dwell time before encryption is 48 hours to several

weeks, providing a window for detection and response.

Threat Overview

INC Ransom emerged in July 2023 as a closed RaaS platform, later transitioning to a more

open affiliate model. The group operates a TOR-based leak site and negotiation portal (inc-

decrypt[.]onion) for double extortion. Affiliates are provided Windows and Linux/ESXi

encryptors that append the .INC extension and drop INC-README.TXT and INC-

README.HTML ransom notes in every encrypted directory.

In March 2024, an operator known as "salfetka" listed INC ransomware source code

(Windows + Linux/ESXi) for sale at $300,000 on hacking forums. Unit 42 analysis

subsequently found significant code overlap between INC and the Lynx ransomware family

(48% overall Windows, 91% Linux), confirming Lynx as a derivative. Microsoft Threat

Intelligence has attributed a subset of INC operations to Vanilla Tempest (formerly Vice

Society/DEV-0832), particularly campaigns targeting US healthcare organizations using

Gootloader for initial access.

In January 2026, security firm Cyber Centaurs discovered an operational security failure in

INC infrastructure, recovering data for 12 US organizations from reused Restic backup

repositories that had not been properly secured.

KEY CHARACTERISTICS

First Seen: July 2023

Type: RaaS (double extortion)

Aliases: GOLD IONIC, G1032

Affiliate: Vanilla Tempest

Platforms: Windows, Linux, ESXi

Extension: .INC

Victims (2025): 300+

Derivative: Lynx ransomware

Dwell Time: 48 hrs to weeks

2 MITRE ATT&CK Mapping

Based on MITRE ATT&CK Group G1032 and Software S1139. All mappings verified against the official ATT&CK matrix.

T1190

Exploit Public-Facing Application

CVE-2023-3519 (Citrix NetScaler), CVE-2023-

48788 (Fortinet EMS), FG-IR-24-535 (FortiOS)

T1566

Phishing / SEO Poisoning

Gootloader delivery via SEO-poisoned search

results (Vanilla Tempest campaigns)

T1003

OS Credential Dumping

lsassy.py for LSASS memory dumping, Impacket

secretsdump.py, HackTool.PS1.VeeamCreds

T1558.003

Kerberoasting

Kerberoasting for Domain Admin credential

acquisition (Source: HvS-Consulting)

T1219

Remote Access Software

AnyDesk, ScreenConnect, TightVNC deployed

for persistent remote access

T1021.001

Remote Desktop Protocol

RDP for lateral movement between

compromised systems (mstsc.exe)

T1570

Lateral Tool Transfer

PsExec used for ransomware deployment across

network shares

T1018

Remote System Discovery

netscan.exe, Advanced IP Scanner, net.exe

group/domain enumeration

T1482

Domain Trust Discovery

nltest.exe /dclist, AdFind.exe for Active Directory

reconnaissance

T1562.001

Disable or Modify Tools

ProcessHacker,

HackTool.Win32.ProcTerminator.A to kill security

processes

T1567.002

Exfiltration to Cloud Storage

MEGASync, Rclone with include.txt target lists,

Restic backup utility

T1053.005

Scheduled Task

INC_Update scheduled task for persistence

(Source: ANY.RUN)

T1486

Data Encrypted for Impact

.INC extension, INC-README.TXT/HTML notes,

Restart Manager API for file release

T1490

Inhibit System Recovery

vssadmin.exe / WMIC shadow copy deletion

before encryption

3 Attack Chain

INC Ransomware Kill Chain

Typical attack progression from initial access to impact

PHASE 1: INITIAL ACCESS

↓

PHASE 2: CREDENTIAL HARVEST & DISCOVERY

↓

PHASE 3: LATERAL MOVEMENT & PERSISTENCE

↓

PHASE 4: DATA EXFILTRATION

↓

PHASE 5: RANSOMWARE DEPLOYMENT & IMPACT

4 Hunt Modules

1 Initial Access — Exploited Public-Facing Applications

Objective: Identify exploitation attempts or post-exploitation activity targeting Citrix NetScaler ADC/Gateway and Fortinet products, the primary

initial access vectors used by INC Ransom affiliates.

Background

INC operators consistently exploit internet-facing network appliances for initial access. Secureworks and MOXFIVE documented CVE-2023-

3519 (Citrix NetScaler RCE) as a primary entry point. ReliaQuest observed CVE-2023-48788 (Fortinet EMS SQL injection), and HvS-Consulting

documented exploitation of FortiGate FortiOS (FG-IR-24-535), where attackers created new SSL VPN configurations to maintain access.

Hunt Procedure

Operational Note

If you operate Citrix NetScaler ADC/Gateway or Fortinet products, verify patches for CVE-2023-3519, CVE-2023-4966 (Citrix Bleed), CVE-2023-

48788, and FG-IR-24-535. Check for unauthorized SSL VPN configurations or admin accounts created after the vulnerability disclosure dates.

2 Credential Access — Credential Dumping & Kerberoasting

Objective: Detect credential harvesting activity using tools and techniques associated with INC Ransom: lsassy.py, Impacket secretsdump.py,

VeeamCreds PowerShell scripts, and Kerberoasting attacks.

Background

Credential theft is a critical early-stage activity for INC operators. Huntress documented the use of lsassy.py for LSASS memory dumping.

ReliaQuest observed Impacket's secretsdump.py and pass-the-hash attacks. HvS-Consulting documented Kerberoasting to obtain Domain

Admin credentials, and Trend Micro flagged HackTool.PS1.VeeamCreds for harvesting credentials from Veeam backup infrastructure.

Hunt Procedure

High-Confidence Indicator

Detection of lsassy.py or secretsdump.py execution is a strong indicator of active compromise. Combined with LSASS access patterns (GrantedAccess

0x1010 or 0x1038), this warrants immediate investigation and host isolation.

3 Unauthorized Remote Management Tools

Objective: Identify installations of AnyDesk, ScreenConnect, TightVNC, or PuTTY on systems where these tools are not authorized, indicating

potential INC Ransom affiliate activity.

Background

INC operators routinely install legitimate remote access tools for persistent, interactive access to compromised networks. Microsoft documented

AnyDesk deployment in Vanilla Tempest campaigns. Huntress observed AnyDesk and TightVNC installations. ReliaQuest noted AnyDesk

installation immediately after CVE-2023-48788 exploitation. These tools provide operators with GUI access that bypasses typical command-line

logging.

Hunt Procedure

Filtering Guidance

Cross-reference results against your authorized RMM tool inventory. Legitimate AnyDesk or ScreenConnect installations on IT admin workstations

should be excluded. Focus on installations on servers, domain controllers, or workstations outside your standard remote support tooling.

4 Network Reconnaissance & Discovery

Objective: Detect execution of network scanning and Active Directory enumeration tools consistently observed in INC Ransom intrusions:

netscan.exe, AdFind, Advanced IP Scanner, nltest.exe, and net.exe domain commands.

Background

After establishing initial access and credentials, INC operators conduct extensive network reconnaissance. Huntress documented the use of

nltest.exe, net.exe, and AdFind for Active Directory enumeration. Secureworks observed netscan.exe for network scanning. These tools help

operators identify high-value targets (domain controllers, file servers, backup infrastructure, Hyper-V hosts) before lateral movement.

Hunt Procedure

Operational Note

Individual use of net.exe or nltest.exe by system administrators is normal. Focus on rapid sequential execution of multiple reconnaissance tools within a

short window (5-10 minutes), execution from non-admin workstations, or execution by service accounts that should not be running interactive

commands.

5 Defense Evasion — Security Tool Termination

Objective: Detect attempts to disable or terminate endpoint security tools using ProcessHacker, ProcTerminator, or similar utilities, a pre-

encryption step consistently observed in INC Ransom attacks.

Background

Before deploying the encryptor, INC operators disable security tools to prevent detection and blocking. Trend Micro documented

HackTool.Win32.ProcTerminator.A, a tool specifically designed to kill security processes. ProcessHacker is also used for the same purpose. This

activity typically occurs in the final hours before encryption and is a strong indicator that ransomware deployment is imminent.

Hunt Procedure

Immediate Escalation Required

Detection of security tool termination tools (ProcessHacker, ProcTerminator) or unexpected security service stops on production systems is a critical

indicator that ransomware deployment may be imminent. Escalate immediately to your incident response team and consider proactive network

isolation of the affected host.

6 Data Staging & Exfiltration

Objective: Identify data staging with 7-Zip/WinRAR and exfiltration activity using MEGASync, Rclone, or Restic — the primary exfiltration tools

observed in INC Ransom operations.

Background

INC operators exfiltrate data as part of the double-extortion model. Trend Micro documented 7-Zip and WinRAR for staging, with MEGASync and

Rclone for cloud exfiltration. Huntress observed Rclone used with an include.txt file specifying targeted directories. ReliaQuest documented

Restic backup utility disguised as a Windows update binary for covert exfiltration, placed in the System32 directory. Some INC attacks are

exfiltration-only, with no encryption deployed.

Hunt Procedure

Operational Note

Rclone and Restic are legitimate backup tools. Focus on execution on systems where these tools are not part of your standard backup infrastructure,

execution by non-service accounts, or connection to cloud storage providers (MEGA, Backblaze, etc.) that are not part of your approved vendor list.

7 Lateral Movement — RDP & PsExec

Objective: Detect anomalous RDP connections and PsExec usage consistent with INC Ransom lateral movement patterns, including pass-the-

hash attacks and ransomware staging via PsExec.

Background

INC operators use RDP extensively for lateral movement after obtaining credentials. HvS-Consulting documented RDP as the primary lateral

movement method. PsExec is used both for interactive command execution and for ransomware binary deployment to multiple hosts

simultaneously. Secureworks observed PsExec being used specifically for final ransomware deployment across the network. ReliaQuest

documented pass-the-hash attacks via Impacket's wmiexec.py for WMI-based lateral movement.

Hunt Procedure

Baseline Guidance

RDP is heavily used in most environments. Establish a baseline of normal RDP patterns (which accounts, between which hosts, at what times) and hunt

for deviations. Focus on RDP from non-admin accounts, server-to-server RDP, RDP during off-hours, or RDP originating from recently compromised

hosts.

8 Persistence Mechanisms

Objective: Detect INC-specific persistence mechanisms including the INC_Update scheduled task, registry modifications via

SystemSettingsAdminFlows.exe, and unauthorized local account creation.

Background

ANY.RUN analysis documented the INC_Update scheduled task as a persistence mechanism. Additionally, INC operators have been observed

modifying system settings via SystemSettingsAdminFlows.exe and creating local administrator accounts for backup access. Combined with the

RMM tools detected in Module 3, these mechanisms ensure operators can return to the network even if individual access vectors are discovered

and remediated.

Hunt Procedure

Operational Note

The INC_Update scheduled task name is a known IOC. However, operators may use different task names in future campaigns. Focus also on scheduled

tasks created by non-standard accounts, tasks that execute binaries from unusual locations (Temp, AppData, Public directories), or tasks created on

multiple hosts within a short timeframe.

9 ESXi & Hypervisor Targeting

Objective: Detect indicators that INC operators are targeting ESXi or Hyper-V hypervisor infrastructure for maximum impact encryption, a

hallmark of INC's Linux variant operations.

Background

INC Ransomware includes both Windows and Linux/ESXi variants. MOXFIVE documented ESXi targeting in incident response engagements.

HvS-Consulting reported ransomware deployment on Hyper-V servers within 48 hours of initial compromise. Encrypting hypervisor

infrastructure causes maximum disruption by taking down all hosted virtual machines simultaneously, making this a high-priority target for INC

operators.

Hunt Procedure

Critical Infrastructure Alert

Any unauthorized access to ESXi or Hyper-V management interfaces should be treated as a critical finding. INC operators have been observed

encrypting hypervisor infrastructure within 48 hours of initial compromise. If suspicious activity is detected on hypervisor hosts, immediately verify

VM snapshots and backup integrity.

10 Ransomware Execution Artifacts

Objective: Detect active or historical INC ransomware execution indicators including ransom note creation, .INC file extension changes, shadow

copy deletion, and known binary indicators.

Background

INC ransomware appends the .INC extension to encrypted files and drops both INC-README.TXT and INC-README.HTML ransom notes in

every directory. The encryptor uses the Windows Restart Manager API (RmStartSession, RmRegisterResources) to release files locked by

running processes before encrypting them. Shadow copies are deleted using vssadmin.exe or WMIC to prevent recovery. Deployment is typically

performed via PsExec across multiple hosts or via WMI Provider Host (Vanilla Tempest). HvS-Consulting identified the binary name win.exe in

their incident response.

Hunt Procedure

Immediate Escalation Required

Detection of INC-README ransom notes, .INC file extensions, or shadow copy deletion indicates ransomware execution is either in progress or has

already occurred. This requires immediate incident response escalation: isolate affected hosts from the network, preserve forensic evidence, and

engage your incident response team or retainer. Do not attempt to recover or decrypt files without IR guidance.

A Appendix A: Indicators of Compromise

Source Information

All IOCs sourced from Trend Micro, Secureworks, Huntress, HvS-Consulting, AlienVault OTX, and ANY.RUN. Last verified: February 11, 2026.

File Indicators

Type Value Context

SHA256 63e0d4e861048f581c9e5c64b28a053eb0023d58eebf2b943868d5f68a67a8b7 Linux ELF variant (Source: Trend Micro)

SHA256 a0ceb258924ef004fa4efeef4bc0a86012afdb858e855ed14f1bbd31ca2e42f5 Linux ELF variant (Source: Trend Micro)

SHA256 c41ab33986921c812c51e7a86bd3fd0691f5bba925fae612f1b717afaa2fe0ef Linux ELF variant (Source: Trend Micro)

SHA256 3dd6fab5ec9444ef7d2f4d2a744b3a4efa1420e346b47efae34145a2aa3b7508 Windows variant (Source: AlienVault OTX)

Extension .INC Encrypted file extension

Ransom Note INC-README.TXT Dropped in every directory during encryption

Ransom Note INC-README.HTML Dropped in every directory during encryption

PDB Path C:\source\INC Encryptor\Release\INC Encryptor.pdb Windows variant debug path (Source: Huntress)

Executable win.exe Ransomware binary (Source: HvS-Consulting)

Executable p443x64.exe Meterpreter dropper (Source: Secureworks)

Config File include.txt Rclone exfiltration target list (Source: Huntress)

Scheduled Task INC_Update Persistence mechanism (Source: ANY.RUN)

Network Indicators

Type Value Context

TOR Domain inc-decrypt[.]onion Payment/negotiation portal (Source: ANY.RUN)

Behavioral Indicators

Behavior Detection Logic

Mass file extension change to .INC File modify events with .INC extension at high volume (>10 per host)

Ransom note deployment INC-README.TXT/HTML creation in multiple directories

Shadow copy deletion vssadmin.exe or WMIC shadow delete commands

RMM tool on servers AnyDesk/ScreenConnect/TightVNC on server-class hosts

Reconnaissance tools netscan.exe, AdFind.exe execution on non-admin workstations

Credential dumping lsassy.py, secretsdump.py process creation or LSASS memory access

Data exfiltration MEGASync, Rclone, Restic on server-class hosts

Scheduled task persistence INC_Update task creation (Event ID 4698)

Important Note on Hash-Based Detection

Because each INC locker may contain a victim-specific ID, file hashes are not reliably reusable across incidents. Prioritize behavioral indicators for detection

rather than relying solely on hash matching. (Source: ANY.RUN)

B Appendix B: Detection Rules

INC Ransomware Note Detection SIGMA

INC File Encryption Activity SIGMA

Shadow Copy Deletion (Ransomware Context) SIGMA

INC Exfiltration Tool Detection SIGMA

INC Persistence Task Detection SIGMA

INC Reconnaissance Tool Detection SIGMA

INC Credential Dumping Detection SIGMA

Detection Resources

MITRE ATT&CK Enterprise Matrix — https://attack.mitre.org/matrices/enterprise/

CISA Known Exploited Vulnerabilities (KEV) Catalog — https://www.cisa.gov/known-exploited-vulnerabilities-catalog

Intruvent Technologies | TLP�CLEAR INT-THG-2026-INC-RANSOM-v1.0

Hunting for INC Ransom / GOLD IONIC Ransomware-as-a-Service Operations

Intruvent Technologies INT-THG-2026-INC-RANSOM-v1.0

Exploit Citrix NetScaler CVE-2023-3519 / CVE-2023-4966 (Citrix Bleed)•

Exploit Fortinet EMS CVE-2023-48788 or FortiOS FG-IR-24-535•

Gootloader via SEO poisoning (Vanilla Tempest campaigns)•

Deploy Meterpreter dropper (p443x64.exe) or Supper backdoor•

Credential dumping via lsassy.py, secretsdump.py, VeeamCreds PowerShell•

Kerberoasting for Domain Admin credentials•

Network scanning with netscan.exe, AdFind, Advanced IP Scanner•

Domain trust enumeration via nltest.exe, net.exe•

Install AnyDesk, ScreenConnect, TightVNC for persistent access•

RDP lateral movement with harvested credentials•

PsExec deployment to domain-joined systems•

Create INC_Update scheduled task for persistence•

Stage data with 7-Zip / WinRAR for archival•

Exfiltrate via MEGASync, Rclone (with include.txt targeting), or Restic•

10,000+ files typically exfiltrated per incident•

Some attacks are exfiltration-only (no encryption deployed)•

Kill security processes with ProcessHacker / ProcTerminator•

Delete shadow copies via vssadmin.exe / WMIC•

Deploy encryptor via PsExec or WMI Provider Host•

Files encrypted with .INC extension; INC-README.TXT/HTML dropped•

Target Hyper-V / ESXi servers for maximum impact•

Duration: 30 min Priority: Critical

                        
# Hunt 1a: Citrix NetScaler exploitation indicators
# Look for web shell creation or suspicious process spawning from Citrix services
index=web OR index=proxy sourcetype=citrix*
| search (uri_path="*gateway*" OR uri_path="*vpn*")
  AND (status>=500 OR http_method="POST")
| stats count by src_ip, uri_path, status
| where count > 20

# Hunt 1b: Fortinet exploitation indicators
# Search for FortiClient EMS SQL injection patterns
index=network sourcetype=fortinet*
| search (event_type="utm" OR event_type="attack")
  AND (attack_name="*SQL*" OR attack_name="*injection*")
| stats count by src_ip, dest_ip, attack_name

# Hunt 1c: New SSL VPN configuration (FG-IR-24-535 post-exploit)
# Unexpected SSL VPN portal or user creation on FortiGate
index=network sourcetype=fortinet*
| search action="config" AND (config_object="vpn.ssl.web.portal" OR config_object="user.local")
| stats count by admin_user, config_object, action
                    

SIEM / EDR QUERY

Duration: 30 min Priority: Critical

                        
# Hunt 2a: LSASS memory access (credential dumping)
index=edr sourcetype=sysmon EventCode=10
| search TargetImage="*\\lsass.exe"
  AND (GrantedAccess="0x1010" OR GrantedAccess="0x1038" OR GrantedAccess="0x1FFFFF")
  AND NOT SourceImage IN ("*\\MsMpEng.exe", "*\\csrss.exe", "*\\svchost.exe")
| stats count by SourceImage, TargetImage, GrantedAccess, host

# Hunt 2b: Python-based credential tools (lsassy, secretsdump)
index=edr (process_name="python.exe" OR process_name="python3.exe")
| search (command_line="*lsassy*" OR command_line="*secretsdump*"
          OR command_line="*wmiexec*" OR command_line="*impacket*")
| stats count by host, user, command_line

# Hunt 2c: Kerberoasting detection
# Windows Security Event 4769 with encryption type 0x17 (RC4)
index=wineventlog EventCode=4769 Ticket_Encryption_Type=0x17
| search Service_Name!="krbtgt" AND Service_Name!="*$"
| stats count by Account_Name, Service_Name, Client_Address
| where count > 5

# Hunt 2d: Veeam credential harvesting
index=edr (process_name="powershell.exe" OR process_name="pwsh.exe")
| search (command_line="*VeeamCreds*" OR command_line="*Get-VBRCredentials*"
          OR command_line="*Veeam.Backup*")
| stats count by host, user, command_line
                    

SIEM / EDR QUERY

Duration: 20 min Priority: High

                        
# Hunt 3a: RMM tool process execution on non-approved hosts
index=edr sourcetype=process_creation
| search process_name IN (
    "AnyDesk.exe", "anydesk.exe",
    "ScreenConnect.Client.exe", "ScreenConnect.WindowsClient.exe",
    "tvnserver.exe", "tvnviewer.exe",
    "putty.exe", "plink.exe"
  )
| stats count earliest(_time) as first_seen latest(_time) as last_seen by host, process_name, user
| eval first_seen=strftime(first_seen, "%Y-%m-%d %H:%M:%S")
| eval last_seen=strftime(last_seen, "%Y-%m-%d %H:%M:%S")

# Hunt 3b: RMM tool installation artifacts
index=edr sourcetype=file_creation
| search (file_path="*\\AnyDesk*" OR file_path="*\\ScreenConnect*"
          OR file_path="*\\TightVNC*")
  AND file_path="*Program Files*"
| stats count by host, file_path

# Hunt 3c: RMM service creation
index=wineventlog EventCode=7045
| search (Service_Name="*AnyDesk*" OR Service_Name="*ScreenConnect*"
          OR Service_Name="*tvnserver*")
| stats count by host, Service_Name, Service_File_Name
                    

SIEM / EDR QUERY

Duration: 25 min Priority: High

                        
# Hunt 4a: Network scanning tool execution
index=edr sourcetype=process_creation
| search process_name IN ("netscan.exe", "NETSCAN.EXE",
    "Advanced_IP_Scanner.exe", "advanced_ip_scanner.exe",
    "AdFind.exe", "adfind.exe")
| stats count by host, user, process_name, parent_process_name

# Hunt 4b: Active Directory enumeration commands
index=edr sourcetype=process_creation
| search (
    (process_name="nltest.exe" AND (command_line="*/dclist*" OR command_line="*/domain_trusts*"))
    OR
    (process_name="net.exe" AND (command_line="*group*" OR command_line="*user*"
      OR command_line="*domain*" OR command_line="*/dom*"))
    OR
    (process_name="ESENTUTL.exe" AND command_line="*/y*" AND command_line="*ntds*")
  )
| stats count by host, user, process_name, command_line

# Hunt 4c: Rapid successive enumeration (behavioral pattern)
index=edr sourcetype=process_creation
| search process_name IN ("net.exe", "nltest.exe", "whoami.exe", "ipconfig.exe", "systeminfo.exe")
| bucket _time span=5m
| stats dc(process_name) as unique_tools count by host, user, _time
| where unique_tools >= 3
                    

SIEM / EDR QUERY

Duration: 20 min Priority: Critical

                        
# Hunt 5a: Process termination tool execution
index=edr sourcetype=process_creation
| search process_name IN ("ProcessHacker.exe", "processhacker.exe",
    "ProcTerminator.exe", "procterminator.exe",
    "PCHunter.exe", "pchunter.exe",
    "KillAV.exe")
| stats count by host, user, process_name, parent_process_name

# Hunt 5b: Security process termination (Event-based)
index=edr
| search (event_type="process_terminate" OR event_type="process_end")
  AND process_name IN (
    "MsMpEng.exe", "MsSense.exe", "SentinelAgent.exe",
    "CylanceSvc.exe", "CSFalconService.exe", "cb.exe",
    "SophosAgent.exe", "bdagent.exe"
  )
| stats count by host, process_name, parent_process_name
| where count > 0

# Hunt 5c: Tamper protection alerts or service stops
index=wineventlog EventCode=7036
| search (Service_Name="*Defender*" OR Service_Name="*Endpoint*"
          OR Service_Name="*Sentinel*" OR Service_Name="*CrowdStrike*")
  AND Service_State="stopped"
| stats count by host, Service_Name
                    

SIEM / EDR QUERY

Duration: 30 min Priority: Critical

                        
# Hunt 6a: Cloud exfiltration tool execution
index=edr sourcetype=process_creation
| search process_name IN ("MEGASync.exe", "megasync.exe",
    "MEGASyncSetup64.exe",
    "rclone.exe", "restic.exe")
| stats count by host, user, process_name, command_line

# Hunt 6b: Rclone with configuration or inclusion files
index=edr sourcetype=process_creation process_name="rclone.exe"
| search (command_line="*--include-from*" OR command_line="*include.txt*"
          OR command_line="*--config*" OR command_line="*copy*"
          OR command_line="*sync*")
| stats count by host, user, command_line

# Hunt 6c: Data staging with archival tools on servers
index=edr sourcetype=process_creation
| search process_name IN ("7z.exe", "7za.exe", "rar.exe", "WinRAR.exe")
  AND (command_line="*a *" OR command_line="*-p*")
| stats count by host, user, command_line
| where count > 3

# Hunt 6d: Restic masquerading in System32
index=edr sourcetype=file_creation
| search file_path="*\\System32\\*"
  AND (file_name="restic.exe" OR file_name="restic*"
       OR file_hash_sha256 IN ("known_restic_hashes"))
| stats count by host, file_path, file_name

# Hunt 6e: Large outbound data transfers
index=network sourcetype=firewall
| search bytes_out > 100000000
  AND dest_port IN (443, 8443, 80)
| stats sum(bytes_out) as total_bytes by src_ip, dest_ip
| eval total_gb=round(total_bytes/1073741824, 2)
| where total_gb > 1
| sort -total_gb
                    

SIEM / EDR QUERY

Duration: 25 min Priority: High

                        
# Hunt 7a: Anomalous RDP logons (Type 10)
index=wineventlog EventCode=4624 Logon_Type=10
| stats count dc(dest) as unique_targets by Account_Name, Source_Network_Address
| where unique_targets > 3 OR count > 10
| sort -unique_targets

# Hunt 7b: PsExec service installation and execution
index=wineventlog EventCode=7045
| search Service_Name="PSEXESVC"
| stats count by host, Service_File_Name, Account_Name

# Hunt 7c: PsExec process execution patterns
index=edr sourcetype=process_creation
| search (process_name="PsExec.exe" OR process_name="psexec.exe"
          OR process_name="PSEXESVC.exe" OR process_name="psexesvc.exe")
| stats count by host, user, process_name, command_line

# Hunt 7d: WMI-based lateral movement (wmiexec)
index=edr sourcetype=process_creation
| search parent_process_name="WmiPrvSE.exe"
  AND (process_name="cmd.exe" OR process_name="powershell.exe")
| stats count by host, user, process_name, command_line
| where count > 0

# Hunt 7e: Pass-the-hash detection (NTLM Type 3 logon)
index=wineventlog EventCode=4624 Logon_Type=3
  Authentication_Package="NTLM"
| stats count dc(dest) as targets by Account_Name, Source_Network_Address
| where targets > 3
                    

SIEM / EDR QUERY

Duration: 20 min Priority: High

                        
# Hunt 8a: INC_Update scheduled task creation
index=wineventlog EventCode=4698
| search TaskName="*INC*"
| stats count by host, TaskName, SubjectUserName

# Hunt 8b: Suspicious scheduled task creation (broader)
index=edr sourcetype=process_creation process_name="schtasks.exe"
| search command_line="*/create*"
  AND NOT command_line IN ("*Microsoft*", "*Windows*", "*Adobe*")
| stats count by host, user, command_line

# Hunt 8c: SystemSettingsAdminFlows registry modification
index=edr sourcetype=process_creation
| search process_name="SystemSettingsAdminFlows.exe"
| stats count by host, user, parent_process_name, command_line

# Hunt 8d: Local account creation (potential backup access)
index=wineventlog EventCode=4720
| stats count by TargetUserName, SubjectUserName, host
| where SubjectUserName != "SYSTEM"

# Hunt 8e: New VPN configurations on network appliances
# Check FortiGate/Citrix admin logs for SSL VPN portal creation
index=network sourcetype=fortinet*
| search action="add" AND config_object="vpn*"
| stats count by admin_user, config_object, host
                    

SIEM / EDR QUERY

Duration: 25 min Priority: Critical

                        
# Hunt 9a: Suspicious SSH access to ESXi hosts
# Review ESXi shell access logs for unauthorized connections
index=esxi OR index=syslog sourcetype=vmware*
| search (event_type="ssh" OR message="*sshd*")
  AND (action="accepted" OR message="*Accepted*")
| stats count by src_ip, user, host
| where user != "root" OR count > 5

# Hunt 9b: ESXi shell enablement
index=esxi OR index=syslog sourcetype=vmware*
| search message="*Shell*" AND (message="*enable*" OR message="*start*")
| stats count by host, message

# Hunt 9c: VM shutdown commands (pre-encryption)
index=esxi OR index=syslog sourcetype=vmware*
| search (message="*vim-cmd vmsvc/power.off*"
          OR message="*esxcli vm process kill*"
          OR message="*stop*" AND message="*vm*")
| stats count by host, message
                    

ESXI / SSH LOGS

                        
# Hunt 9d: Hyper-V VM management from unusual sources
index=edr sourcetype=process_creation
| search (process_name="powershell.exe" OR process_name="pwsh.exe")
  AND (command_line="*Stop-VM*" OR command_line="*Get-VM*"
       OR command_line="*Hyper-V*")
| stats count by host, user, command_line

# Hunt 9e: File transfer to ESXi/Hyper-V hosts
index=network sourcetype=firewall
| search dest_port=22 AND dest_ip IN (YOUR_HYPERVISOR_IPS)
| stats count sum(bytes_out) as total_bytes by src_ip, dest_ip
| where total_bytes > 10000000
                    

HYPER-V / WINDOWS

Duration: 15 min Priority: Critical

                        
# Hunt 10a: Ransom note creation
index=edr sourcetype=file_event
| search file_name IN ("INC-README.TXT", "INC-README.HTML")
  AND event_type="file_create"
| stats count by host

# Hunt 10b: Mass file encryption indicator (.INC extension)
index=edr sourcetype=file_event
| search file_name="*.INC" AND event_type="file_modify"
| stats count by host
| where count > 10

# Hunt 10c: Shadow copy deletion
index=edr sourcetype=process_creation
| search (
    (process_name="vssadmin.exe" AND command_line="*delete*" AND command_line="*shadow*")
    OR
    (process_name="wmic.exe" AND command_line="*shadowcopy*" AND command_line="*delete*")
  )
| stats count by host, process_name, command_line

# Hunt 10d: PsExec deployment pattern (ransomware distribution)
index=edr sourcetype=process_creation
| search (process_name="PsExec.exe" OR process_name="PSEXESVC.exe")
| stats count by host, src
| where count > 5

# Hunt 10e: Mass service stop commands (pre-encryption)
index=edr sourcetype=process_creation
| search process_name="net.exe" AND command_line="*stop*"
| stats count by host
| where count > 5

# Hunt 10f: Known INC binary indicators
index=edr sourcetype=process_creation
| search (
    process_name="win.exe"
    OR original_file_name="INC Encryptor*"
    OR file_description="INC Encryptor*"
  )
| stats count by host, process_name, file_hash
                    

SIEM / EDR QUERY

title: INC Ransomware Note File Creation
status: experimental
description: Detects creation of INC-README.TXT or INC-README.HTML ransom notes
logsource:
    category: file_event
    product: windows
detection:
    selection:
        EventType: creation
        TargetFilename|endswith:
            - '\INC-README.TXT'
            - '\INC-README.HTML'
    condition: selection
level: critical
tags:
    - attack.impact
    - attack.t1486

title: Mass File Extension Change to .INC
status: experimental
description: Detects mass file modification with .INC extension indicating active encryption
logsource:
    category: file_event
    product: windows
detection:
    selection:
        EventType: modification
        TargetFilename|endswith: '.INC'
    condition: selection | count() by host > 10
    timeframe: 5m
level: critical
tags:
    - attack.impact
    - attack.t1486

title: Shadow Copy Deletion - Ransomware Pre-Encryption
status: experimental
description: Detects shadow copy deletion commonly used by INC before encryption
logsource:
    category: process_creation
    product: windows
detection:
    selection_vssadmin:
        Image|endswith: '\vssadmin.exe'
        CommandLine|contains|all:
            - 'delete'
            - 'shadows'
    selection_wmic:
        Image|endswith: '\wmic.exe'
        CommandLine|contains|all:
            - 'shadowcopy'
            - 'delete'
    condition: selection_vssadmin or selection_wmic
level: high
tags:
    - attack.impact
    - attack.t1490

title: Cloud Exfiltration Tools on Server Hosts
status: experimental
description: Detects MEGASync, Rclone, or Restic execution on server-class systems
logsource:
    category: process_creation
    product: windows
detection:
    selection:
        Image|endswith:
            - '\MEGASync.exe'
            - '\MEGASyncSetup64.exe'
            - '\rclone.exe'
            - '\restic.exe'
    condition: selection
level: high
tags:
    - attack.exfiltration
    - attack.t1567.002

title: INC_Update Scheduled Task Creation
status: experimental
description: Detects creation of the INC_Update scheduled task used for persistence
logsource:
    category: process_creation
    product: windows
detection:
    selection_schtasks:
        Image|endswith: '\schtasks.exe'
        CommandLine|contains: 'INC_Update'
    selection_eventlog:
        EventID: 4698
        TaskName|contains: 'INC'
    condition: selection_schtasks or selection_eventlog
level: high
tags:
    - attack.persistence
    - attack.t1053.005

title: Network Reconnaissance Tools - INC Ransom Toolset
status: experimental
description: Detects execution of netscan.exe and AdFind.exe used by INC operators
logsource:
    category: process_creation
    product: windows
detection:
    selection:
        Image|endswith:
            - '\netscan.exe'
            - '\NETSCAN.EXE'
            - '\AdFind.exe'
            - '\Advanced_IP_Scanner.exe'
    condition: selection
level: high
tags:
    - attack.discovery
    - attack.t1046
    - attack.t1018

title: Credential Dumping Tools - INC Ransom Toolset
status: experimental
description: Detects lsassy.py and secretsdump.py execution used by INC for credential theft
logsource:
    category: process_creation
    product: windows
detection:
    selection_python:
        Image|endswith: '\python.exe'
        CommandLine|contains:
            - 'lsassy'
            - 'secretsdump'
    selection_lsass_access:
        EventID: 10
        TargetImage|endswith: '\lsass.exe'
        GrantedAccess|contains:
            - '0x1010'
            - '0x1038'
    condition: selection_python or selection_lsass_access
level: critical
tags:
    - attack.credential_access
    - attack.t1003

